
Quelques bugs célèbres

1940 Mark-I : une erreur de calcul est détectée. Une inspection
de la machine montre un court-circuit dû à un insecte (bug)

≈ 1985 Appareil de radio-thérapie Therac-25.
Erreur de dosage dans des traitements du cancer (morts)

≈ 1990 Processeur arithmétique de l’Intel Pentium II.
≈ 475M$ pour remplacer les processeurs

1995 Système de gestion des bagages à l’aéroport de Denver.
Retard d’ouverture de 9 mois, ≈ 1M$/jour

1996 Explosion d’Ariane 5 lors de son vol inaugural.
Utilisation d’un contrôleur d’Ariane 4 ne supportant pas les
données d’accélérations d’Ariane 5 (dépassement de
capacité).

Cycle de développement d’un logiciel

Plus un bug est détecté tôt, moins il a d’impact ! 1

1. Extrait de Principles of Model Checking, C. Baier and J-P Katoen, 2008
Nicolas Bedon - Introduction au model-checking - 4/134



Détecter les bugs
Quelques approches complémentaires :
▶ revue de code

▶ le canard en plastique
▶ tests

▶ simulation de scénarios
▶ peut être réalisée

- sur la spécification
- sur la réalisation
- sur le tout
- sur une partie

▶ automatiques ou non
▶ tests unitaires

▶ locaux
▶ peuvent être orthogonaux à l’application
▶ automatiques ou non

▶ tests généraux
▶ plate-formes d’exécution outillées (ex : valgrind)

▶ model-checking
Les tests doivent être refaits après modifications pour détecter
les régressions.

Nicolas Bedon - Introduction au model-checking - 5/134



Tester vs vérifier

Test : à partir d’une donnée,
un seul chemin est testé.

Vérification : tous les chemins
sont testés !

Vie du système et model-checking

Expression des besoins

Analyse

Design

Code

Tests

Maintenance

Model-checking

Principe du model-checking
Besoins

Spécification
de propriété

Système

Système
modélisé

Model-checking

Propriété vérifiée

Propriété
non vérifiée :

contre-exemple
Simulation

Localisation
de l’erreur

Formalisation Modélisation

Deux approches du model-checking

Top→bottom Bottom→top

Design

Modèle abstrait
pour vérification

Code

Model-
checker

Modèle abstrait
pour vérification

Code

Abstractions

Techniques de raffinement Abstractions

Quelques outils populaires pour le model-checking

▶ Spin : vérification des systèmes distribués
▶ Promela
▶ LTL
▶ http://www.spinroot.com

▶ SMV, NuSVM
▶ LTL, CTL, . . .
▶ http://nusmv.fbk.eu/

▶ UPPAAL : vérification des systèmes temps-réels
▶ Automates temporisés
▶ http://www.uppaal.org

▶ Java PathFinder
▶ JVM spécialisée
▶ Vérification de programmes Java
▶ https://babelfish.arc.nasa.gov/trac/jpf/

▶ et beaucoup d’autres !

Systèmes réactifs

▶ Réagit à l’environnement ;
▶ Ne termine pas.

Entrée Emetteur Canal Récepteur Sortie

msg/seq#

ack/seq#

▶ Le canal peut supprimer ou dupliquer des paquets ;
▶ L’émetteur retransmet ses messages (avec intervalle)

jusqu’à ce que l’ACK correspondant soit reçu ;
▶ Les numéros de séquences sont modulo 2 (bit alternant).

Chaque composant (émetteur, récepteur, canal) est un
système réactif à lui seul.
Le temps est une notion centrale pour les propriétés du
système.

Vocabulaire

Grandes classes de propriétés en model-checking :
▶ sécurité (safety ) : rien de mauvais n’arrivera jamais

Exemple : x est toujours inférieur à k
Algorithme : trouver un chemin du système vers un état
dans lequel le mauvais arrive

▶ vivacité (liveness) : quelque chose de positif finira par
arriver
Exemple : le système arrivera dans un état final
Algorithme : trouver une boucle (infinie) dans laquelle la
chose positive n’arrive pas

▶ divers

Simple Promela INterpreter

▶ Open-source LTL Model-checking spinroot.com

▶ Interfaces graphiques : ispin, xspin, jspin
Spin est axé sur les systèmes décrits par interactions de
processus.
Communication entre processus :
▶ primitives de rendez-vous (synchrones) ;
▶ échange de messages asynchrones par canaux

bufferisés ;
▶ variables partagées.

Simple Promela INterpreter

Plusieurs modes de fonctionnement :
1. Émulation :

▶ choix pseudo-aléatoire d’exécution
▶ simulation guidée par l’utilisateur

2. Vérification
2.1 de progression du système
2.2 d’états de terminaison invalides
2.3 d’assertions
2.4 d’une propriété P (LTL).
Sortie :
▶ OK si tout va bien
▶ un exemple de configuration du système violant la propriété

sinon.

PROcess MEta LAnguage : description de modèle
Composée de :
▶ des définitions de constantes entières (enum du C) :

mtype ...
▶ des définitions de types :

typedef ...
▶ des définitions de canaux de communication ;

chan ch = [capacité] of {type, ...}
▶ asynchrone : capacité > 0
▶ rendez-vous (synchrone) : capacité = 0

▶ des définitions de variables globales ;
▶ des définitions de processus ;

proctype
▶ variables locales
▶ instructions

▶ un processus d’initialisation optionnel.
init

Promela : principales instructions
Instructions de base :
▶ affectation (toujours exécutable) ;
▶ instruction à valeur (exécutable si non zéro (ex : true)) ;
▶ envoyer (ch !) (exécutable si ch n’est pas plein) ;
▶ recevoir (ch?) (exécutable si ch n’est pas vide) ;
▶ assertion (assert(expr)) (toujours exécutable) ;
▶ affichage (printf) (toujours exécutable) ;

Instructions à valeur :
▶ skip (toujours exécutable, équivaut à 1 ou true)
▶ timeout (exécution variable, true si aucune autre

instruction n’est exécutable) ;
Instructions composées :
▶ if (exécutable si au moins une garde l’est)
▶ do (exécutable si au moins une garde l’est)
▶ atomic... (exécutable si la première instruction l’est)
▶ d_step... (exécutable si la première instruction l’est)

Instructions de contrôle de flot :
▶ goto (saut à une instruction étiquetée) ;
▶ break (sortie de do).

Utiliser Spin
L’exécutable spin :
▶ permet la simulation à partir d’un fichier PROMELA

spin exemple.pml
La simulation peut avoir différentes caractéristiques (cf
options de simulation)

▶ permet la génération de code C pour un vérificateur
généré à partir d’un fichier PROMELA
spin -a exemple.pml
▶ génère pan.c
▶ la compilation de pan.c peut être paramétrée par des

options de compilation
▶ la vérification peut être paramétrée par des options de

l’exécutable
▶ si le vérificateur trouve un problème, il génère un

contre-exemple (exemple.pml.trail) qui peut être
rejoué par spin
spin -t exemple.pml

Tout ceci peut être enveloppé dans une interface graphique
(d’utilisation optionnelle) : ispin, xspin, jspinNicolas Bedon - Introduction au model-checking - 17/134



Utiliser Spin
Plusieurs modes de fonctionnement :

1. Émulation :
▶ choix pseudo-aléatoire d’exécution

spin exemple.pml
▶ simulation guidée par l’utilisateur

spin -i exemple.pml
▶ exécution d’une trace

spin -t exemple.pml

2. Vérification
spin -a exemple.pml
2.1 de progression du système

gcc -O3 -DNP pan.c && a.out -b -l
2.2 d’états de terminaison invalides

réalisé par défaut par l’exécutable
2.3 d’assertions

réalisé par défaut par l’exécutable
2.4 d’une propriété P (LTL)

gcc -O3 pan.c && a.out -a -b

Spin : premier exemple
Producteurs/consommateurs

#define NP 2 /* #prod */
#define NC 3 /* #cons */

mtype = { P, C };

mtype turn = P;

active [NP] proctype producer()
{

do
:: (turn == P) ->

printf("Produce%d\n", _pid);
turn = C

od
}
active [NC] proctype consumer()
{

do
:: (turn == C) ->

printf("Consume%d\n", _pid);
turn = P

od
}

spin prodcons.pml

Produce1
Consume2

Consume3
Produce0

Produce1
Consume3

Produce0
Produce1

Consume2
Consume3

Produce0
Produce1

Consume3
Produce0

Consume2
Produce1

Produce0
Consume3

Produce0
Produce1

Consume2
...

▶ La simulation est pseudo-aléatoire
▶ Elle peut également être interactive (option -i)
▶ L’option -u permet de spécifier un nombre de pas à simuler.

Spin : algorithme de simulation
Chaque processus est transformé en automate.

4 2

3

turn==P

printf("Produce

%d\n",_pid);turn=C

4 2

3

turn==C

printf("Consume

%d\n",_pid);turn=P

▶ Les états peuvent avoir plusieurs types :
▶ normal (rien de spécial)
▶ progress (marque un état de progrès - cycles de non

progression)
▶ end (terminaison du processus)
▶ accept (marque un état pour la détection de cycles

d’acceptation)
▶ Les transitions peuvent être atomiques ou non
▶ Afficher les automates : option -d du vérificateur

Spin : algorithme de simulation

Tant que tous les processus ne sont pas bloqués
Calculer la liste des actions qui peuvent être exécutées
(contient la prochaine action à exécuter de chaque
processus non bloqué)

Si elle n’est pas vide
Choisir une action a au hasard
(ou demander un choix à l’utilisateur en cas de
simulation interactive)

L’exécuter ; si elle génère une erreur la propager

s ta

Figure – L’exécution de a fait passer le système de l’état s vers t .

Logiques temporelles

Logiques modales : la valeur de vérité des propositions change
par rapport à une modalité.

Logiques temporelles = modales sur le temps.

Exemple : opérateurs de changements temporels : Demain,
UnJour

▶ Proposition P : il fait beau
▶ Maintenant, il fait beau : P
▶ Demain, il fera beau : DemainP
▶ Un jour il ne fera plus beau : UnJour¬P

LTL

Linear Temporal Logic (Pnueli 1977)
▶ Le temps est organisé comme N (temps discret)
▶ Pas de variables, donc pas de quantificateurs.
▶ Les atomes sont les propositions
▶ Les connecteurs logiques habituels ¬,∧,∨ sont autorisés
▶ Des modificateurs modaux peuvent s’appliquer aux

formules
Exemples :
▶ Maintenant il fait beau : P
▶ Un jour il ne fera plus beau : ⋄¬P
▶ Il fera beau tous les jours : □P
▶ Les changements de temps sont infinis :
□((P → ⋄¬P) ∧ ((¬P) → ⋄P))

LTL : syntaxe

Temps :

t0 t1 t2 t3 t4

Soit P un ensemble fini de propositions.

Formules atomiques : P ∈ P

Modificateurs modaux :
▶ ◦ϕ : à la prochaine unité de temps, ϕ est vraie ;
▶ □ϕ : à partir de maintenant et pour toujours, ϕ est vraie ;
▶ ⋄ϕ : ϕ est vraie maintenant, ou sera vraie dans le futur ;
▶ ϕUψ ; ϕ est vraie jusqu’à ce que ψ le devienne.

Connecteurs logiques : ¬, ∧, ∨.

LTL : sémantique
À chaque unité de temps, un ensemble de propositions sont
vraies (les autres sont fausses)
¬Q Q Q Q ¬Q
P ¬P ¬P ¬P P

t0 t1 t2 t3 t4

u = u0u1u2 . . . ,
ui = {P : P vraie au temps ti} ∪ {¬P : P fausse au temps ti}

Sémantique :
▶ u |= P si P ∈ u0 ;
▶ u |= ◦ϕ si u[1[|= ϕ ;
▶ u |= □ϕ si u[i[|= ϕ pour tout i ∈ N ;
▶ u |= ⋄ϕ s’il existe i ∈ N tel que u[i[|= ϕ ;
▶ u |= ϕUψ s’il existe i ∈ N tel que u[i[|= ψ, et u[j[|= ϕ pour

tout j ∈ 0 . . . i − 1.

LTL : remarques
Remarques :
▶ ◦ est parfois noté X ou N (NeXt)
▶ □ est parfois noté G (Globally)
▶ ⋄ est parfois noté F (Finally)
▶ □ϕ ≡ ¬ ⋄ ¬ϕ
▶ ⋄ϕ ≡ ¬□¬ϕ ≡ trueUϕ
▶ (□ϕ) ∧ (□ϕ′) ≡ □(ϕ ∧ ϕ′)
▶ (⋄ϕ) ∨ (⋄ϕ′) ≡ ⋄(ϕ ∨ ϕ′)
▶ (□ϕ) ∨ (□ϕ′) ̸≡ □(ϕ ∨ ϕ′)
▶ (⋄ϕ) ∧ (⋄ϕ′) ̸≡ ⋄(ϕ ∧ ϕ′)

Extensions :
▶ Until faible : ϕWϕ′ ≡ (□ϕ) ∨ (ϕUϕ′)
▶ Opérateurs pour le passé : ♦(P),■(H),S sont les versions

« passé » de ⋄,□,U .
▶ Ajouter ces opérateurs n’augmente pas l’expressivité de

LTL.

Syntaxe LTL de Spin

ψ ::=true | false | proposition
[]ψ | <> ψ | Xψ | !ψ

ψUψ′ | ψ&&ψ′ | ψ||ψ′ | ψ− > ψ′ | ψ < − > ψ′ | (ψ)

Spin : premier exemple

#define NP 2 /* #prod */
#define NC 3 /* #cons */

mtype = { P, C };

mtype turn = P;

ltl { []((turn==P) -> <>(turn==C)) }

active [NP] proctype producer()
{

do
:: (turn == P) ->

printf("Produce%d\n", _pid);
turn = C

od
}
active [NC] proctype consumer()
{

do
:: (turn == C) ->

printf("Consume%d\n", _pid);
turn = P

od
}

$ spin -a prodcons.pml
ltl ltl_0: [] ((! ((turn==P)))

|| (<> ((turn==C))))
$ gcc -O3 pan.c
$ ./a.out -a -b
(Spin Version 6.4.7 -- 19 August 2017)
+ Partial Order Reduction

Full statespace search for:
never claim + (ltl_0)
assertion violations + (if within scope of claim)
acceptance cycles + (fairness disabled)
invalid end states - (disabled by never claim)

State-vector 60 byte, depth reached 201, errors:0
252 states, stored (330 visited)
529 states, matched
859 transitions (= visited+matched)

0 atomic steps
hash conflicts: 72 (resolved)

Stats on memory usage (in Megabytes):
0.021 equivalent memory usage for states (stored*(State-vector + overhead))
0.290 actual memory usage for states

128.000 memory used for hash table (-w24)
0.534 memory used for DFS stack (-m10000)

128.730 total actual memory usage

unreached in proctype producer
prodcons.pml:17, state 7, "-end-"
(1 of 7 states)
unreached in proctype consumer
prodcons.pml:25, state 7, "-end-"
(1 of 7 states)
unreached in claim ltl_0
_spin_nvr.tmp:10, state 13, "-end-"
(1 of 13 states)

pan: elapsed time 0 secondsNicolas Bedon - Introduction au model-checking - 28/134



Spin : premier exemple

128.000 memory used for hash table (-w24)
0.534 memory used for DFS stack (-m10000)

128.730 total actual memory usage

unreached in proctype producer
prodcons.pml:17, state 7, "-end-"
(1 of 7 states)
unreached in proctype consumer
prodcons.pml:25, state 7, "-end-"
(1 of 7 states)
unreached in claim ltl_0
_spin_nvr.tmp:10, state 13, "-end-"
(1 of 13 states)

pan: elapsed time 0 seconds

La taille d’un seul état

Nicolas Bedon - Introduction au model-checking - 28/134



Spin : premier exemple

#define NP 2 /* #prod */
#define NC 3 /* #cons */

mtype = { P, C };

mtype turn = P;

ltl { []((turn==P) -> <>(turn==C)) }

active [NP] proctype producer()
{

do
:: (turn == P) ->

printf("Produce%d\n", _pid);
turn = C

od
}
active [NC] proctype consumer()
{

do
:: (turn == C) ->

printf("Consume%d\n", _pid);
turn = P

od
}

$ spin -a prodcons.pml
ltl ltl_0: [] ((! ((turn==P)))

|| (<> ((turn==C))))
$ gcc -O3 pan.c
$ ./a.out -a -b
(Spin Version 6.4.7 -- 19 August 2017)
+ Partial Order Reduction

Full statespace search for:
never claim + (ltl_0)
assertion violations + (if within scope of claim)
acceptance cycles + (fairness disabled)
invalid end states - (disabled by never claim)

State-vector 60 byte, depth reached 201, errors:0
252 states, stored (330 visited)
529 states, matched
859 transitions (= visited+matched)

0 atomic steps
hash conflicts: 72 (resolved)

Stats on memory usage (in Megabytes):
0.021 equivalent memory usage for states (stored*(State-vector + overhead))
0.290 actual memory usage for states

128.000 memory used for hash table (-w24)
0.534 memory used for DFS stack (-m10000)

128.730 total actual memory usage

unreached in proctype producer
prodcons.pml:17, state 7, "-end-"
(1 of 7 states)
unreached in proctype consumer
prodcons.pml:25, state 7, "-end-"
(1 of 7 states)
unreached in claim ltl_0
_spin_nvr.tmp:10, state 13, "-end-"
(1 of 13 states)

pan: elapsed time 0 seconds

Plus grande exécution

Nicolas Bedon - Introduction au model-checking - 28/134



Spin : premier exemple

#define NP 2 /* #prod */
#define NC 3 /* #cons */

mtype = { P, C };

mtype turn = P;

ltl { []((turn==P) -> <>(turn==C)) }

active [NP] proctype producer()
{

do
:: (turn == P) ->

printf("Produce%d\n", _pid);
turn = C

od
}
active [NC] proctype consumer()
{

do
:: (turn == C) ->

printf("Consume%d\n", _pid);
turn = P

od
}

$ spin -a prodcons.pml
ltl ltl_0: [] ((! ((turn==P)))

|| (<> ((turn==C))))
$ gcc -O3 pan.c
$ ./a.out -a -b
(Spin Version 6.4.7 -- 19 August 2017)
+ Partial Order Reduction

Full statespace search for:
never claim + (ltl_0)
assertion violations + (if within scope of claim)
acceptance cycles + (fairness disabled)
invalid end states - (disabled by never claim)

State-vector 60 byte, depth reached 201, errors:0
252 states, stored (330 visited)
529 states, matched
859 transitions (= visited+matched)

0 atomic steps
hash conflicts: 72 (resolved)

Stats on memory usage (in Megabytes):
0.021 equivalent memory usage for states (stored*(State-vector + overhead))
0.290 actual memory usage for states

128.000 memory used for hash table (-w24)
0.534 memory used for DFS stack (-m10000)

128.730 total actual memory usage

unreached in proctype producer
prodcons.pml:17, state 7, "-end-"
(1 of 7 states)
unreached in proctype consumer
prodcons.pml:25, state 7, "-end-"
(1 of 7 states)
unreached in claim ltl_0
_spin_nvr.tmp:10, state 13, "-end-"
(1 of 13 states)

pan: elapsed time 0 seconds

La taille d’un seul état

Plus grande exécution

#erreurs trouvées

0.290 actual memory usage for states

128.000 memory used for hash table (-w24)
0.534 memory used for DFS stack (-m10000)

128.730 total actual memory usage

unreached in proctype producer
prodcons.pml:17, state 7, "-end-"
(1 of 7 states)
unreached in proctype consumer
prodcons.pml:25, state 7, "-end-"
(1 of 7 states)
unreached in claim ltl_0
_spin_nvr.tmp:10, state 13, "-end-"
(1 of 13 states)

pan: elapsed time 0 seconds

La taille d’un seul état

Plus grande exécution

#erreurs trouvées

#états

Mémoire totale

Spin : premier exemple

#define NP 2 /* #prod */
#define NC 3 /* #cons */

mtype = { P, C };

mtype turn = P;

ltl { []((turn==P) -> !<>(turn==C)) }

active [NP] proctype producer()
{

do
:: (turn == P) ->

printf("Produce%d\n", _pid);
turn = C

od
}
active [NC] proctype consumer()
{

do
:: (turn == C) ->

printf("Consume%d\n", _pid);
turn = P

od
}

$ spin -a prodcons.pml
ltl ltl_0: [] ((! ((turn==P)))

|| (!<> ((turn==C))))
$ gcc -O3 pan.c

$ ./a.out -a -b
pan:1: assertion violated !((turn==1)) (at depth 6)
pan: wrote prodcons.pml.trail

(Spin Version 6.4.7 -- 19 August 2017)
Warning: Search not completed
+ Partial Order Reduction

Full statespace search for:
never claim + (ltl_0)
assertion violations + (if within scope of claim)
acceptance cycles + (fairness disabled)
invalid end states - (disabled by never claim)

State-vector 60 byte, depth reached 6, errors: 1
4 states, stored
0 states, matched
4 transitions (= stored+matched)
0 atomic steps

hash conflicts: 0 (resolved)

Stats on memory usage (in Megabytes):
0.000 equivalent memory usage for states (stored*(State-vector + overhead))
0.289 actual memory usage for states

128.000 memory used for hash table (-w24)
0.534 memory used for DFS stack (-m10000)

128.730 total actual memory usage

pan: elapsed time 0 seconds

Spin : premier exemple
Simulation du contre-exemple :

$ spin -t prodcons.pml
ltl ltl_0: [] ((! ((turn==P))) || (! (<> ((turn==C)))))
Never claim moves to line 4 [(!(!((turn==P))))]
Never claim moves to line 10 [(1)]

Produce1
spin: _spin_nvr.tmp:9, Error: assertion violated
spin: text of failed assertion: assert(!((turn==C)))
Never claim moves to line 9 [assert(!((turn==C)))]
spin: trail ends after 7 steps
#processes: 5

turn = C
7: proc 4 (consumer:1) prodcons.pml:20 (state 4)
7: proc 3 (consumer:1) prodcons.pml:20 (state 4)
7: proc 2 (consumer:1) prodcons.pml:20 (state 4)
7: proc 1 (producer:1) prodcons.pml:12 (state 4)
7: proc 0 (producer:1) prodcons.pml:12 (state 4)
7: proc - (ltl_0:1) _spin_nvr.tmp:8 (state 16)

5 processes created

Spin : premier exemple
Simulation du contre-exemple avec davantage de détails :
$ spin -t -v prodcons.pml
ltl ltl_0: [] ((! ((turn==P))) || (! (<> ((turn==C)))))
starting claim 2
using statement merging

1: proc - (ltl_0:1) _spin_nvr.tmp:4 (state 4) [(!(!((turn==P))))]
Never claim moves to line 4 [(!(!((turn==P))))]

2: proc 1 (producer:1) prodcons.pml:13 (state 1) [((turn==P))]
3: proc - (ltl_0:1) _spin_nvr.tmp:10 (state 14) [(1)]

Never claim moves to line 10 [(1)]
Produce1

4: proc 1 (producer:1) prodcons.pml:14 (state 2) [printf(’Produce%d\\n’,_pid)]
5: proc - (ltl_0:1) _spin_nvr.tmp:10 (state 14) [(1)]
6: proc 1 (producer:1) prodcons.pml:15 (state 3) [turn = C]
7: proc - (ltl_0:1) _spin_nvr.tmp:9 (state 11) [((turn==C))] <merge 0 now @12>

spin: _spin_nvr.tmp:9, Error: assertion violated
spin: text of failed assertion: assert(!((turn==C)))

7: proc - (ltl_0:1) _spin_nvr.tmp:9 (state 12) [assert(!((turn==C)))]
Never claim moves to line 9 [assert(!((turn==C)))]
spin: trail ends after 7 steps
#processes: 5

turn = C
7: proc 4 (consumer:1) prodcons.pml:20 (state 4)
7: proc 3 (consumer:1) prodcons.pml:20 (state 4)
7: proc 2 (consumer:1) prodcons.pml:20 (state 4)
7: proc 1 (producer:1) prodcons.pml:12 (state 4)
7: proc 0 (producer:1) prodcons.pml:12 (state 4)
7: proc - (ltl_0:1) _spin_nvr.tmp:8 (state 16)

5 processes created

L’interface graphique Ispin

L’interface graphique Ispin

L’interface graphique Ispin

L’interface graphique Ispin

Spin : algorithme de vérification

Spin transforme également les formules de LTL en automates

□((turn == P) → ¬ ⋄ (turn == C))

8

2 16

12

((!(!((turn==P)))
&&(turn==C))) (!(!((turn==P))))

(true)

assert
(!((!(!((turn==P)))
&&(turn==C))))

((turn==C))

(true)

assert(!((turn==C)))

Spin : algorithme de vérification
p1 . . . pn

p1 ⊗ · · · ⊗ pn
Automate de Büchi

ϕ

¬ϕ

A¬ϕ
Automate de Büchi

∩

▶ L’intersection doit être vide.
▶ Spin cherche un cycle :

▶ contenant un état acceptant
▶ accessible à partir d’un état initial

▶ Cette recherche est réalisée par un parcours en
profondeur

▶ Les différentes constructions sont réalisées à la volée :
▶ les états ne sont construits que si nécessaire
▶ les construits sont stockés dans une table de hachage

Promela : processus

Un processus est défini par proctype suivi de :
▶ un nom pour le processus
▶ une liste de paramètres
▶ les déclarations de variables locales
▶ les instructions composant le processus

Les processus peuvent être créés à tout moment
▶ par run

proctype p1(int n) { ... }
init {

...
run p1(12);
...

}

▶ création automatique par active devant proctype
active proctype p2(byte b) { ... }

Promela : processus

L’ordonnancement du processus peut être influencé :

bit a;
active [2] proctype p() provided (a==_pid) {
do
:: printf("%d\n",_pid) -> a=(a+1)%2;
od
}

Ici, alternance strict de l’ordonnancement.

Promela : types

Types de base : bit, bool, byte, short, int
Tableaux :
type nom [const ] <= expr >
▶ L’initialisation <= expr > est optionnelle. L’expression expr

est évaluée une fois, puis tous les éléments du tableau
sont initialisés à sa valeur. En cas d’absence, initialisation
à 0 ou équivalent.

▶ Les tableaux de bit ou bool sont stockés comme des
tableaux C d’unsigned char (pas d’économie de place).

▶ t = t ′ : réalise t [0] = t ′[0]
▶ Tableaux multidimensionnels : indirectement en utilisant

typedef

Définition de type :
typedef nom { type1 nom1; . . . ; typen nomn};

Promela : variables

Peuvent être :
▶ globales
▶ locales à un processus :

déclaration au début

Initialisation par défaut : 0 ou équivalent

Changement de valeur :
▶ affectation,
▶ passage d’argument,
▶ passage de message

Les problèmes de typage sont détectés à run-time.

Promela : variables et types
La taille d’un état dépend de l’espace occupé par les données.
▶ Attention à ne pas utiliser trop de variables ;
▶ Utiliser le type le plus adapté.

Promela dispose des types d’entiers suivants :
▶ bit : 0,1

▶ bool : false,true

▶ byte : 0 . . . 255

▶ mtype : 1 . . . 255

▶ short : −215 . . . 215 − 1

▶ int : −231 . . . 231 − 1

▶ unsigned : 0 . . . 232 − 1

Type unsigned : le nombre de bits peut être choisi (impacte le
nombre de valeurs représentables) :

unsigned v : 5; /* unsigned stored in 5 bits */
unsigned w : 3 = 5; /* value range 0..7, initially 5 */

Promela : variables

Variables prédéfinies :
▶ _pid : read-only, locale à chaque processus, contient son

numéro. Les numéros commencent à 0 et sont par ordre
d’exécution ;
init est un processus comme les autres.

▶ _ : write-only et globale : c’est une poubelle ;
▶ _last : globale, numéro du dernier processus à avoir été

exécuté ;
▶ np_ : globale, vrai si et seulement si le système ne

progresse pas (définition à venir...).

Promela : if

if
:: garde1− > instr1,1; . . . ; instr1,n1 ;

. . .

:: gardek− > instrk ,1; . . . ; instrk ,nk ;

:: else− > instrk+1,1; . . . ; instrk+1,nk+1 ;

fi ;

Le else est optionnel. Il est exécutable si aucune des gardes
ne l’est.
L’instruction if est
▶ exécutable si une des gardes ou le else sont exécutables.
▶ bloquée sinon

Promela : do

do
:: garde1− > instr1,1; . . . ; instr1,n1 ;

. . .

:: gardek− > instrk ,1; . . . ; instrk ,nk ;

od ;

La boucle est infinie :
▶ une garde exécutable est élue pseudo-aléatoirement à

chaque tour ;
▶ un break dans les instructions l’interrompt.

Promela : for (depuis SPIN 6)
▶ Itérer sur un intervalle de valeurs

for ’(’ name ’:’ expr ’..’ expr ’)’ ’{’ sequence ’}’

▶ Itérer sur les indices des éléments d’un tableau
for ’(’ name in array ’)’ ’{’ sequence ’}’

▶ Lire un canal (sans en retirer les éléments)
for ’(’ name in channel ’)’ ’{’ sequence ’}’

Exemple (Bien respecter les espaces autour des ..) :

int i;
for (i : 1 .. 10) {

printf("i = %d\n", i)
}

est équivalent à

i = 1; // Toujours exécutable
do
:: i <= 10 -> printf("i = %d\n", i); i++
:: else -> break
od

Promela : select (depuis SPIN 6)

Choisir une valeur dans un intervalle. Toujours exécutable.
Prends plusieurs pas.

Exemple (Bien respecter les espaces autour des ..) :

int i;
select (i : 8 .. 17);

est équivalent à

i = 8; // Toujours exécutable
do
:: i < 17 -> i++
:: break
od

Promela : canaux

FIFO
chan nom = [capacite] of {type1, . . . , typen};

Capacité :
▶ > 0 pour un canal asynchrone ;
▶ 0 pour une communication synchrone par rendez-vous.

type1, . . . , typen sont les types des objets pouvant être mis
dans le canal.

Envoi : canal!expr1, . . . ,exprn
Exécutable si le canal n’est pas plein.

Réception : canal?var1, . . . , varn
Exécutable si le canal n’est pas vide.

Promela : canaux

La réception peut se faire sous la condition de motifs :
canal?var1, . . . ,expri , . . . , varn

La lecture n’est exécutable que si
▶ le canal n’est pas vide
▶ et si le message à lire a la forme fixée par les expri .

Si la lecture est réalisée, les autres parties lues sont stockées
dans les varj .

Si la valeur de varj doit être utilisée comme motif, on remplace
varj par eval(varj).

Exemple : c?1,eval(x), y
Lit le message se trouvant dans c s’il est composé d’un 1, de la
valeur de x et stocke la troisième valeur lue dans y .

Promela : tests sur canaux (hors rendez-vous)

Il est possible de seulement tester l’exécutabilité d’une
opération (hors rendez-vous) :
canal?[var1, . . . , varn]

Ce test peut aussi être réalisé avec l’utilisation de motifs.

L’exécution ne modifie ni le canal ni les vari .

Poll : c? < var1, . . . , varn >
Comme c?var1, . . . , varn mais ne retire pas le message du
canal.

Promela : tests sur canaux de rendez-vous
Technique : ajout de la valeur d’une condition dans le canal.
// Semaphore de Dijkstra

#define P 0 /* Prendre */
#define V 1 /* Vendre (=rendre) */

#define NB 2 /* #max users pouvant détenir le sémaphore simultanément */

/* Canal d’accession au sémaphore.
Le booléen doit toujours être true pour les requêtes.
C’est la valeur d’une expression en interne au sémaphore

*/
chan sema_chan = [0] of { bit, bool };

/* Primitives d’accès au sémaphore */
#define SEMA_P(s) s?P,true
#define SEMA_V(s) s!V,true

/* #user en section critique */
byte nb=0;

active proctype semaphore() {
byte count=NB;
do
:: sema_chan!P,count>0 -> count--;
:: sema_chan?V,true -> count++;
od

}

active [NB+2] proctype user() {
do
::SEMA_P(sema_chan);

/* Section critique */
nb++;
printf("User %d en section critique\n", _pid);
assert(nb<=NB);
nb--;
SEMA_V(sema_chan);
printf("User %d a rendu le semaphore\n", _pid);
/* Sortie de section critique */

od
}

Promela : canaux
Il est possible de mettre un canal dans un autre :

chan g = [1] of { chan };

active proctype p1() {
chan l = [1] of { byte };
byte v;
g!l;
l?v -> printf("%d\n", v)

}

active proctype p2() {
chan w;
g?w;
w!3

}

$ spin chan.pml
3

2 processes created

Promela : canaux

Autres opérations sur les canaux :

▶ len(c) : nombre de messages dans c
▶ empty(c) : équivalent à len(c)==0

À utiliser quand l’atomicité est voulue
▶ nempty(c) : équivalent à len(c)>0

À utiliser quand l’atomicité est voulue (ne pas employer
!empty(c))

▶ full(c) : cf. empty
▶ nfull(c) : cf. nempty

Promela : rendez-vous

Un rendez-vous permet de synchroniser deux processus.
Il est réalisé par l’utilisation d’un canal de capacité 0.

chan canal = [0] of { bit, int };

active proctype p1()
{

/* A chaque tour de boucle,
on sort ou on ne sort pas */

do
:: skip -> skip;
:: skip -> break;
od;
printf("\%d est sorti de sa boucle,

il attends...", _pid);
canal!1,3;
printf("\%d rentre dans

sa boucle infinie!", _pid);
do
:: skip ->
od;

}

active proctype p2()
{

bit b;
int i;
/* A chaque tour de cette boucle,

une chance sur deux de sortir */
do
:: skip -> skip;
:: skip -> break;
od;
printf("\%d est sorti de sa boucle,

il attends...", _pid);
canal?b,i;
printf("\%d rentre dans

sa boucle infinie!", _pid);
do
:: skip ->
od;

}

Promela : instructions atomiques

Les instructions peuvent être regroupées pour être exécutées
de manière atomique. atomic {instr1; . . . ; instrn; }

La suite est exécutable si instr1 l’est.

La suite d’instructions n’est atomique que si toutes sont
exécutables. Si, en cours d’exécution de la suite, une
instruction instri n’est pas exécutable, un autre processus peut
être exécuté, puis, quand instri redevient exécutable,
l’exécution de la suite reprend sans aucun signalement.

Les sauts en dehors de et à l’intérieur du bloc atomique sont
permis.

Promela : instructions atomiques

Une suite déterministe d’instructions est une version plus
contrainte qu’une suite atomique.

d_step {instr1; . . . ; instrn; }

La suite est exécutable si instr1 l’est.

Contrairement à atomic :
▶ le non-déterminisme éventuel à l’intérieur de la suite est

résolu de manière déterministe (mais inconnue) ;
▶ les sauts en dehors et dans la suite sont interdits ;
▶ une instruction bloquante (qui n’est pas la première)

provoque une erreur.

atomic et d_step peuvent être utilisés pour réduire le nombre
d’états du modèle.

Nicolas Bedon - Introduction au model-checking - 56/134



timeout/else

timeout :
▶ variable spéciale de Promela ;
▶ vraie si et seulement si aucune autre instruction du

système n’est ordonnançable ;
▶ attention :

▶ il s’agit donc d’un timeout global au système
▶ rien à voir avec la notion de temps

else :
▶ variable spéciale de Promela ;
▶ pouvant être utilisée dans un if ou un do ;
▶ vraie si et seulement si aucune autre garde du if/do n’est

exécutable.

Techniques de bruitage de canaux
Deux processus emetteur et recepteur communiquent à
travers un canal non bruité.
L’émetteur émet et attend acquittement de son message avant
de transmettre le message suivant ;
active proctype emetteur() {
m=...message initial...
do
:: toR!MSG,m ->

toE?ACK,m;
m=...message suivant...

od

Le récepteur attend un message et l’acquitte à réception avant
d’attendre le message suivant.
active proctype recepteur() {
do
:: toR?MSG,m ->

toE!ACK,m
od

Techniques de bruitage de canaux
Le bruitage du canal peut être introduit par un démon

active proctype bruiteur() {
do
:: toR?_,_
:: toE?_,_
od

Les messages pouvant maintenant être perdus on les
re-émets :

active proctype emetteur() {
m=...message initial...
do
:: toR!MSG,m ->

if
:: toE?ACK,m -> m=...message suivant...
:: timeout /* message perdu */
fi

od

Techniques de bruitage de canaux

L’utilisation d’un processus supplémentaire pour le démon peut
être évitée :
▶ en intégrant la perte de message émis du coté de

l’émetteur :
active proctype emetteur() {
m=...message initial...
do
:: if /* Emission ou perte ? */

:: toR!MSG,m
:: true
fi
if
:: toE?ACK,m -> m=...message suivant...
:: timeout /* message perdu */
fi

od
}

Techniques de bruitage de canaux
L’utilisation d’un processus supplémentaire pour le démon peut
être évitée :
▶ en intégrant la perte d’acquittement du coté de l’émetteur :

active proctype emetteur() {
m=...message initial...
do
:: if /* Emission ou perte ? */

:: toR!MSG,m
:: true
fi
if
:: toE?ACK,m -> m=...message suivant...
:: toE?_,_ /* ACK perdu */
:: timeout /* message perdu */
fi

od
}

▶ inutile de modifier le récepteur.

Spin : divers

▶ Pas de fonctions
▶ Les macros inline sont autorisées

inline echange(x,y) {
byte t; t=x; x=y; y=t;

}

▶ Un seul format de sortie pour printf : %d
▶ Échappement :

{ P } unless { E }
Exécute P sauf si E est exécutable (et dans ce cas c’est E
qui est exécuté)

Vérifier avec Spin : les assertions
C’est le moyen le plus simple :

byte tour;
byte count;
active [2] proctype p() {
do
:: tour==_pid ->

count=count+1;
assert(count==1);

printf("%d\n",_pid);
tour=(tour+1)%2;
count=count-1;

od
}

$ spin -a assert.pml
$ gcc -O3 pan.c
$ ./a.out
pan:1: assertion violated (count==1) (at depth 7)
pan: wrote assert.pml.trail

(Spin Version 6.4.7 -- 19 August 2017)
Warning: Search not completed
+ Partial Order Reduction

Full statespace search for:
never claim - (none specified)
assertion violations +
acceptance cycles - (not selected)
invalid end states +

State-vector 28 byte, depth reached 7, errors: 1
8 states, stored
0 states, matched
8 transitions (= stored+matched)
0 atomic steps

hash conflicts: 0 (resolved)

Stats on memory usage (in Megabytes):
0.000 equivalent memory usage for states

(stored*(State-vector + overhead))
0.292 actual memory usage for states

128.000 memory used for hash table (-w24)
0.534 memory used for DFS stack (-m10000)

128.730 total actual memory usage

Contrairement aux assertions C, Java, . . ., les assertions sont
vérifiées pour toutes les exécutions possibles.

Vérifier avec Spin : les assertions
Exemple : tester l’atomicité « pure »

L’atomicité « pure » peut facilement être vérifiée. Il faut
▶ ajouter une nouvelle variable globale atomique

▶ l’initialiser à false

▶ changer atomic { i1; . . . ; in } par
atomic { i1; atomique = true; . . . ; in; atomique =
false }

▶ ajouter un démon
active proctype verifAtomique() {
do
:: assert(!atomique);
od
}

Assertion fausse =⇒ atomicité brisée (si i1 n’est pas
décomposable).

(optimisation à venir)

Vérifier avec Spin : les fins invalides
Fin normale du système : tous les processus ont terminé.

Fin anormale : il reste au moins un processus vivant, mais dont
la prochaine instruction n’est pas exécutable.

Exemple : interblocage.

Déclarer une localisation comme fin potentielle « normale » :
étiquette end[a-zA-Z0-9_]*

active proctype p() {
printf("Début %d\n",_pid);
end:
do
:: false ->
od
printf("Fin %d\n",_pid);
}

Avec l’option -q à l’exécution : les canaux doivent être vides à la
terminaison du système.

Vérifier avec une affirmation « jamais » (never claim)

Affirmation « jamais » : code, mais pas processus
▶ seulement des instructions sans effet de bord
▶ sur les canaux les poll sont autorisés
▶ des variables et fonctions prédéfinies peuvent être utilisées
▶ états d’acceptation : formalisation d’une condition

d’acceptation de Büchi (il ne faut pas accepter).
Aucune exécution du système ne doit aussi être une exécution
d’une affirmation « jamais » qui termine

never {
do
:: !p -> break
od
}

vérifie que p est toujours vrai.

Vérifier avec une affirmation « jamais » (never claim)

▶ np_ : vrai si et seulement le système ne progresse pas
(définition à venir...) ;

▶ _last : dernier processus à avoir été ordonnancé ;
▶ pc_value(pid) : numéro de l’état dans lequel se trouve

pid ;
▶ enabled(pid) : vrai si pid est ordonnançable ;
▶ procname[pid]@label : vrai si la prochaine instruction

pouvant être exécutée par pid est étiquetée par label .
Spécifiques aux affirmations « jamais »

Les formules de LTL sont traduites par Spin en affirmations
« jamais ».

La traduction s’obtient par l’option -f de spin.

Vérifier avec une affirmation « jamais » (never claim)
Exemple : □(p → pUq)

À chaque instant, si p est vrai alors il l’est jusqu’à ce que que q le
devienne (et q le deviendra)

never {
do
:: p && !q -> break;
:: true
od
accept0: do
:: !q
:: !(p || q) -> break
od
}

▶ Il ne faut pas passer infiniment souvent par l’état d’acceptation ;
▶ Il ne faut pas rejoindre l’état final ;
▶ En cas de non éligibilité la propriété est vérifiée.

Ici, si q est vrai dans la seconde boucle, p → pUq est vraie,
inutile de continuer.

Les traces

Elles permettent de spécifier des propriétés sur les canaux.

La suite des opérations sur les canaux doit être une alternance
stricte de lectures de messages de type a dans q1 et
d’écritures de messages de type b dans q2 :

mtype = { a, b };
trace {
do
:: q1!a; q2?b
od
}

Ne doit pas contenir de non-déterminisme.

Version négative : notrace.

Vérifier une propriété de sûreté □P
Technique 1 : créer un processus qui vérifie P en permanence :
active proctype verifierP()
{

do
:: assert(P)
od

}

Peut être optimisé (très efficace !) par
active proctype verifierP()
{

do
:: atomic { !P -> assert(P); }
od

}

pour éviter que assert(P) ne soit exécutable en permanence.
Remarque : la boucle n’est pas nécessaire puisqu’il suffit que
!P soit vérifiée une seule fois dans la vie du système et que
verifierP peut n’être activé qu’à ce moment là par

Vérifier une propriété de sûreté □P
Technique 2 : utiliser never (évite l’utilisation d’un processus) :
never
{

do
:: assert(P)
od

}

Technique 3 :
{ corps de processus }
unless { atomic { !P -> assert(P); } }

▶ à utiliser sur au moins un processus ;
▶ évite aussi l’utilisation d’un processus supplémentaire ;
▶ des optimisations peuvent devenir invalides si le corps du

processus contient un rendez-vous ;
▶ le processus doit ne jamais s’arrêter.

Technique 4 : utiliser LTL
ltl { []P }

Utiliser LTL

Spin transforme la formule en never
▶ les propositions de la formule sont des expressions

booléennes
▶ sans effet de bord ;
▶ pouvant accéder aux variables globales
▶ pouvant accéder aux variables locales

▶ p:x pour la variable x du processus p
▶ p[c]:x pour la variable x de l’instance c du processus p

▶ si empty(), nempty(), full(), nfull() apparaissent
dans une négation, une erreur de syntaxe est générée ;
▶ la négation peut être générée par Spin
▶ les raisons de l’erreur de syntaxe sont liées aux

optimisations de Spin
▶ on peut aussi utiliser len()

ltl { [] (toR?[_,_,0]->(emit:seq2==1)) }

Équité de l’ordonnancement
Équité forte (strong fairness)

Si au temps t la prochaine instruction i du processus P est
exécutable, et qu’elle est exécutable infiniment souvent à partir de t ,
alors P arrivera à exécuter i .

Équité faible (weak fairness)
Si au temps t la prochaine instruction i du processus P est

exécutable, et que l’exécutabilité de i ne change plus, alors P arrivera
à exécuter i .

L’équité faible peut être imposée par l’option -f du vérificateur.
byte n;

ltl { []<>(n==0) }

active [2] proctype p() {
do
:: n=_pid; printf("%d\n",n);
od
}

Erreur sans équité faible, pas d’erreur avec.

Vérifier la progression du système

Définitions :
▶ un processus est dans un état de progression locale s’il est

sur une étiquette progress∈ progress[a-zA-Z0-9_]* ;
▶ le système est dans un état de progression globale si au

moins un des processus actifs est dans un état de
progression locale ;

▶ une exécution est non-progressante si le système ne
passe pas infiniment souvent dans un état de progrès.

La variable globale en lecture seule np_ est vraie si et
seulement si l’état global du système n’est pas un état de
progrès.

Vérifier la progression du système

Spin peut détecter les situations de non-progrès du système
(vérifiant ⋄□np_)
▶ compiler pan.c avec l’option -DNP

▶ intègre au vérificateur le code nécessaire à la détection de
cycle de non-progrès ;

▶ permet d’utiliser l’option -l (détection de cycles de
non-progrès) du vérificateur ;

▶ désactive l’option -a (détection de cycle d’acceptation) du
vérificateur.

▶ exécuter le vérificateur avec l’option -l.

Vérifier la progression du système

active proctype p() {
bool c;
debut:

printf("Début de boucle de %d\n", _pid);
progress: c=true;
printf("Fin de boucle de %d\n", _pid);
c=false;
goto debut;

}

active proctype q() {
bool c;
debut:

printf("Début de boucle de %d\n", _pid);
c=true;
printf("Fin de boucle de %d\n", _pid);
c=false;
goto debut;

}

Pour que le système progresse il faut que p prenne infiniment
souvent la main.
$ spin -a progress2.pml && gcc -O3 -DNP pan.c && ./a.out -b -l
pan:1: non-progress cycle (at depth 2)
...

Vérifier la progression du système

$ spin -t -p progress2.pml
...

Début de boucle de 1
2: proc 1 (q:1) progress2.pml:16 (state 1) [printf(’Début de boucle de %d\\n’,_pid)]
<<<<<START OF CYCLE>>>>>
4: proc 1 (q:1) progress2.pml:17 (state 2) [c = 1]

Fin de boucle de 1
4: proc 1 (q:1) progress2.pml:18 (state 3) [printf(’Fin de boucle de %d\\n’,_pid)]
4: proc 1 (q:1) progress2.pml:19 (state 4) [c = 0]

Début de boucle de 1
6: proc 1 (q:1) progress2.pml:16 (state 1) [printf(’Début de boucle de %d\\n’,_pid)]

spin: trail ends after 6 steps
#processes: 2

6: proc 1 (q:1) progress2.pml:17 (state 2)
6: proc 0 (p:1) progress2.pml:6 (state 1)

Quelques options utiles
▶ Spin

▶ -a : générer le vérificateur
▶ -i : simulation interactive
▶ -t : simuler à partir d’une trace unique
▶ -tN : simuler à partir de la trace N

▶ Compilateur
▶ -DNP : intègre le code de détection de cycles de

non-progrès
▶ permet -l à l’exécution
▶ inhibe -a à l’exécution

▶ Vérificateur
▶ -a : trouver les cycles d’acceptation
▶ -c0 : continuer à trouver des erreurs
▶ -d : afficher un automate (penser à supprimer les

optimisations d’automate de Spin : options -o...)
▶ -e : une trace par erreur
▶ -l : trouver les cycles de non-progrès
▶ -m : augmenter la taille de la pile

(error : max search depth too small = vérification incomplète (voir aussi option -b) !)

Exclusion mutuelle : l’algorithme de Peterson
Deux processus P0 et P1 partagent une ressource, veulent y accéder mais il
est interdit que les deux le fassent simultanément.
Deux variables globales :

bool demande[2] = { false, false }
int tour;

Code de Pi (i ∈ {0; 1}) :

Boucle infinie:
// Début code non critique
// ...
// Fin code non critique
demande[i] = true;
tour = (i+1)%2; // Politesse: on laisse la priorité à l’autre
tant que (demande[(i+1)%2] && tour==(i+1)%2) {
// C’est le tour de l’autre, j’attends

}
// C’est mon tour
// Début section critique
// ... Accès à la ressource ...
// Fin section critique
demande[i] = false

Exclusion mutuelle : l’algorithme de Peterson
bool demande[2];
bit tour;

active [2] proctype p() {
bool critique; /* Le processus est dans sa section critique */
debut:

printf("Section non critique de %d\n", _pid);
demande[_pid] = true;
printf("%d veut prendre la main\n",_pid);
tour = (_pid+1)%2;
do
:: !(demande[(_pid+1)%2] && tour==((_pid+1)%2)) -> break;
od
critique=true;
printf("Section critique de %d\n",_pid);
critique=false;
demande[_pid] = false;
printf("Fin section critique de %d\n",_pid);

goto debut;
}

Exclusion mutuelle : l’algorithme de Peterson

Vérifier l’exclusion mutuelle :

Par LTL
▶ □¬(p[0] : critique ∧ p[1] : critique)
▶

spin -a peter.pml && gcc -O3 pan.c && ./a.out -a -b

Tout va bien.

Par assertions : une variable count globale et

critique=true; count++; assert (count==1);
printf("Section critique de %d\n",_pid);
count--; critique=false;

spin -a peter.pml && gcc -O3 pan.c && ./a.out
Tout va bien.

Exclusion mutuelle : l’algorithme de Peterson
Vérifier l’absence d’interblocage :
(définition : pas d’interblocage si, quand un des deux processus veut rentrer
dans sa section critique, alors un des deux finit par y arriver)
Deux méthodes

▶ par LTL :

▶ □(¬ ⋄□(demande[0] ∧ demande[1]))
▶ spin -a peter.pml && gcc -O3 pan.c && ./a.out -a

Tout va bien.

▶ par test de progrès : on progresse à chaque fois que P0 ou P1 arrive
dans sa section critique.

▶ progress: printf("Section critique de %d\n",_pid);

▶ spin -a peter.pml && gcc -O3 -DNP pan.c && ./a.out -b -l

Tout va bien.

Il faut préférer la seconde méthode :
▶ économie de l’automate de la propriété LTL
▶ l’espace des états à explorer est plus petit

Exclusion mutuelle : l’algorithme de Peterson
Vérifier l’absence de famine :
(définition : un processus meurt de faim s’il n’arrive jamais dans sa section
critique (manger))
Par LTL :
▶ □((demande[0] → ⋄¬demande[0]) ∧ (demande[1] → ⋄¬demande[1]))
▶ spin -a peter.pml && gcc -O3 pan.c && ./a.out -a -b

Erreur !

$ spin -t peter.pml
...

Fin section critique de 0
Section non critique de 0

<<<<<START OF CYCLE>>>>>
Section critique de 1
Fin section critique de 1
Section non critique de 1
1 veut prendre la main

...

La famine est provoquée par l’absence d’équité dans l’ordonnancement.
En ajoutant l’équité faible :
spin -a peter.pml && gcc -O3 pan.c && ./a.out -a -b -f

Pathfinder (1997)

▶ Bug « inversion de priorité »
▶ tâche de priorité basse : récolte des données météo,

publication sur le bus de données
▶ tâche de priorité moyenne : communications
▶ tâche de priorité haute : gestion du bus de données
▶ « inversion » de priorité : la tâche de priorité basse

empêche l’exécution de celle de haute priorité
▶ détecté sur Terre avant lancement, mais correction non

prioritaire
▶ 4 jours de fonctionnement anormal du rover Sojourner (sur

une mission d’un mois)
▶ analyse, détection et correction sur Terre
▶ mise à jour du logiciel du rover

Pathfinder (1997)

Modélisation du système
▶ un verrou assure l’exclusion mutuelle à l’accès au bus ;
▶ deux processus

▶ haut : priorité haute
▶ bas : priorité basse, ne peut s’exécuter que si le processus

de priorité haute est dans l’état INACTIF
▶ les états des processus sont

▶ INACTIF : ne fait rien, ne possède pas le verrou
▶ ATTENTE : en attente du verrou
▶ ACTIF : possède le verrou, travaille sur le bus

et le cycle des états d’un processus est
INACTIF→ATTENTE→ACTIF→INACTIF

Pathfinder (1997)

mtype { LIBRE, OCCUPE };
mtype { INACTIF, ENATTENTE, ACTIF };

mtype mutex;

mtype etat[2];

proctype haut(int n) {
do
:: true ->

etat[n]=ENATTENTE;
atomic { mutex==LIBRE ->

mutex=OCCUPE; }
etat[n]=ACTIF;
atomic { etat[n]=INACTIF;

mutex=LIBRE; }
od
}

proctype bas(int n) provided (etat[1]==INACTIF) {
do
:: true ->

etat[n]=ENATTENTE;
atomic { mutex==LIBRE ->

mutex=OCCUPE; }
etat[n]=ACTIF;
atomic { etat[n]=INACTIF;

mutex=LIBRE; }
od
}

init {
etat[0] = INACTIF;
etat[1] = INACTIF;
mutex=LIBRE;
run haut(1);
run bas(0);
}

$ spin -a pathfinder1.pml && gcc -O3 pan.c && ./a.out
pan:1: invalid end state (at depth 17)
...
$ spin -t -v pathfinder1.pml
...

mutex = OCCUPE
etat[0] = ENATTENTE
etat[1] = ENATTENTE

...

Le processus bas perd la

main juste après avoir pris

le verrou, et il ne peut plus

s’exécuter.

Pathfinder (1997)
Il est aussi possible que même avec un ordonnancement
équitable (weak fair ), le processus bas ne prenne jamais le
verrou :

proctype bas(int n) provided (etat[1]==INACTIF) {
do
:: true ->

etat[n]=ENATTENTE;
atomic { mutex==LIBRE ->

mutex=OCCUPE; }
progress: etat[n]=ACTIF;
atomic { etat[n]=INACTIF;

mutex=LIBRE; }
od
}

$ spin -a pathfinder1.pml
&& gcc -O3 -DNP pan.c && ./a.out -b -l -f

pan:1: non-progress cycle (at depth 58)
...

Pathfinder (1997)

Il est aussi possible que même avec un ordonnancement
équitable (weak fair ), le processus bas ne prenne jamais le
verrou :
$ spin -t -p pathfinder1.pml
...
32: proc 2 (haut:1) pathfinder1.pml:17 (state 8) [mutex = LIBRE]
34: proc 1 (bas:1) pathfinder1.pml:25 (state 2) [etat[n] = ENATTENTE]
36: proc 2 (haut:1) pathfinder1.pml:13 (state 3) [((mutex==LIBRE))]
37: proc 2 (haut:1) pathfinder1.pml:14 (state 4) [mutex = OCCUPE]
39: proc 2 (haut:1) pathfinder1.pml:15 (state 6) [etat[n] = ACTIF]
41: proc 2 (haut:1) pathfinder1.pml:16 (state 7) [etat[n] = INACTIF]
42: proc 2 (haut:1) pathfinder1.pml:17 (state 8) [mutex = LIBRE]
<<<<<START OF CYCLE>>>>>

44: proc 2 (haut:1) pathfinder1.pml:13 (state 3) [((mutex==LIBRE))]
45: proc 2 (haut:1) pathfinder1.pml:14 (state 4) [mutex = OCCUPE]
47: proc 2 (haut:1) pathfinder1.pml:15 (state 6) [etat[n] = ACTIF]
49: proc 2 (haut:1) pathfinder1.pml:16 (state 7) [etat[n] = INACTIF]
50: proc 2 (haut:1) pathfinder1.pml:17 (state 8) [mutex = LIBRE]
...

Conclusion sur Spin
La qualité de la vérification dépend de celle du modèle.
▶ réalisation manuelle du modèle
▶ extraction automatique du modèle à partir d’un programme

▶ MODEX

En pratique, il est parfois difficile de réaliser la vérification
(explosion du nombre d’états).

Formellement, toutes les propriétés ne peuvent pas être
vérifiées
▶ Exemple : « tout chemin d’exécution a toujours une voie de

sortie »
▶ Solution : autres logiques (CTL, . . .)

Spin n’est pas conçu pour la vérification de propriétés temps
réel.

Computation Tree Logic

Une machine

A

B C

L’arbre de toutes ses exécutions
A

B C

A B C

CTL permet de spécifier des propriétés sur les chemins :
▶ il existe un chemin tel que...
▶ pour tous les chemins, ...
▶ tous les états du chemin vérifient ...
▶ un état du chemin vérifie ...

Computation Tree Logic

Formules d’état :

ϕ : true | x | ϕ1 ∧ ϕ2 | ¬ϕ | ∃φ | ∀φ

Formules de chemin :

φ : ◦ϕ | ϕ1Uϕ2

▶ Les quantificateurs portent sur des formules de chemin ;
▶ Tout opérateur temporel est immédiatement précédé d’un

quantificateur ;
▶ Les formules de CTL sont les formules d’état.

Computation Tree Logic

∀□noir

∀ ⋄ noir

∃ ⋄ noir

∃□noir

▶ ∃ ⋄ p ≡ ∃trueUp
▶ ∀ ⋄ p ≡ ∀trueUp
▶ ∃□p ≡ ¬∀ ⋄ ¬p
▶ ∀□p ≡ ¬∃ ⋄ ¬p

▶ ∃pWp′ ≡
¬∀((p ∧ ¬p′)U(¬p ∧ ¬p′))

▶ ∀pWp′ ≡
¬∃((p ∧ ¬p′)U(¬p ∧ ¬p′))

Remarque :
▶ ∀ ⋄ p ≡ ¬∃□¬p

Computation Tree Logic

Exemples :
▶ Exclusion mutuelle : ∀□(¬crit1 ∨ ¬crit2)
▶ p se produira toujours infiniment souvent : ∀□∀ ⋄ p
▶ Tout feu rouge est précédé (hors situation initiale) de

l’orange : ∀□(¬orange → ∀ ◦ ¬rouge)
▶ Toute question recevra une réponse :

∀□(question → ∀ ⋄ réponse)
▶ ∀□P : exprime la sécurité.
▶ ∃ ⋄ P : exprime l’atteignabilité.

Expressivités de CTL et LTL

Les propriétés LTL suivantes ne s’expriment pas en CTL :
▶ ⋄□p
▶ ⋄(p ∧ ◦p)

Les propriétés CTL suivantes ne s’expriment pas en LTL :
▶ ∀ ⋄ ∀□a
▶ ∀ ⋄ (p ∧ ∀ ◦ p)
▶ ∀□∃ ⋄ p

Théorème
(Clarke et Draghicescu) Soient ϕ une formule de CTL et ϕ la
formule de LTL obtenue en retirant simplement à ϕ tout ses
quantificateurs de chemins. Alors
▶ ϕ ≡ ϕ

▶ ou ϕ n’est équivalente à aucune formule de LTL.

Expressivités de CTL et LTL

La propriété CTL ϕ ≡ ∀ ⋄ ∀□a ne s’exprime pas en LTL.

Suppression des quantificateurs : ϕ ≡ ⋄□a.

Soit le système S suivant :

0start 1 2

{a} {a}

Alors S |= ϕ, mais S ̸|= ϕ car sω
0 ̸|= ϕ.

Expressivités de CTL et LTL : équité
La plupart des propriétés d’équité s’expriment en LTL par
▶ (équité forte) : □ ⋄ p
▶ (équité faible) : ⋄□p

Par exemple, si un processus pi (i ∈ {0,1}) est servi infiniment
souvent alors l’autre aussi

∧i∈{0,1}□ ⋄ pi → □ ⋄ p(i+1)%2

Vérifier une propriété Q sur tous les parcours avec équité forte

(∧i∈{0,1}□ ⋄ pi → □ ⋄ p(i+1)%2) → Q

Si le model-checker LTL dispose d’algorithmes particuliers pour
garantir l’équité il faut les utiliser de préférence :
▶ Exemple : -f pour Spin (équité faible)
▶ C’est plus efficace : le model-checker ne construit plus

l’ensemble de toutes les exécutions possibles, seulement
celles garantissant l’équité.

Expressivités de CTL et LTL : équité
La plupart des propriétés d’équité ne s’expriment pas avec CTL

▶ c’est donc au model-checker d’implanter l’algorithme
d’équité

▶ ça n’est pas toujours possible...

Les expressivités de CTL et LTL sont incomparables :
▶ l’équité faible s’exprime avec LTL mais pas avec CTL
▶ « Il existe toujours une possibilité d’exécution amenant à

un état terminal » s’exprime avec CTL mais pas avec LTL

Conclusion
▶ implantation d’algorithmes pour traiter les cas particuliers

importants dans les model-checker
▶ choisir son model-checker en fonction des propriétés à

vérifier.

UPPAAL

▶ Université d’Uppsala (Suède) et Aalborg (Norvège)
▶ Model-checker à base de TCTL
▶ Vérification des systèmes temps-réel
▶ Interface graphique en Java
▶ Vérificateur en C++
▶ Contient un simulateur (interactif ou pseudo-aléatoire)
▶ À base d’automates temporisés
▶ Implante de nombreuses optimisations
▶ Variables entières
▶ Horloges
▶ Fonctions utilisateur
▶ Structures de données structurées
▶ Canaux

Nicolas Bedon - Introduction au model-checking - 98/134



Exclusion mutuelle : l’algorithme de Peterson
Deux processus P0 et P1 partagent une ressource, veulent y accéder mais il
est interdit que les deux le fassent simultanément.
Deux variables globales :

bool demande[2] = { false, false }
int tour;

Code de Pi (i ∈ {0; 1}) :

Boucle infinie:
// Début code non critique
// ...
// Fin code non critique
demande[i] = true;
tour = (i+1)%2; // Politesse: on laisse la priorité à l’autre
tant que (demande[(i+1)%2] && tour==(i+1)%2) {
// C’est le tour de l’autre, j’attends

}
// C’est mon tour
// Début section critique
// ... Accès à la ressource ...
// Fin section critique
demande[i] = false

Exclusion mutuelle : l’algorithme de Peterson

4 états :

Boucle infinie:
PASSIF (P):
// Début code non critique
// ...
// Fin code non critique
demande[i] = true;
DEMANDE (D):
tour = (i+1)%2; // Politesse: on laisse la priorité à l’autre
ATTENTE (A) :
tant que (demande[(i+1)%2] && tour==(i+1)%2) // C’est le tour de l’autre, j’attends
SECTION CRITIQUE (S):
// C’est mon tour
// Début section critique
// ... Accès à la ressource ...
// Fin section critique
demande[i] = false

Exclusion mutuelle : l’algorithme de Peterson

Automate de Pi (i ∈ {0,1})

Pstart D

AS

demande[i] :=true

tour :=(i+1)%2
!demande[(i+1)%2]

tour==i

demande[i] :=false

▶ Mise à jour : réalisée en suivant la transition
▶ Garde : condition pour pouvoir suivre la transition

Exclusion mutuelle : ∀□¬(S0 ∧ S1)

Exclusion mutuelle : l’algorithme de Peterson

Exclusion mutuelle : l’algorithme de Peterson

Exclusion mutuelle : l’algorithme de Peterson

Exclusion mutuelle : l’algorithme de Peterson

Exclusion mutuelle : l’algorithme de Peterson

Exclusion mutuelle : l’algorithme de Peterson

UPPAAL et CTL
UPPAAL ne supporte qu’un sous-ensemble très restreint de
CTL : pas de formules de chemin imbriquées.
Formules de chemins supportées par UPPAAL :

∀□noir

∃ ⋄ noir

∀ ⋄ noir

∃□noir

noir⇝ rouge
(∀□(noir → ∀ ⋄ rouge))

La formule (d’état) deadlock est vraie pour tous les états

UPPAAL et le temps

Le temps est continu.

Le modèle peut utiliser des horloges
▶ comme gardes sur les transitions
▶ comme invariants sur les états

Les horloges peuvent être
▶ testées dans les gardes
▶ (ré-)initialisées dans les mises à jour

K
x ≥ 2
x := 0

UPPAAL et le temps

Le temps est continu.

Le modèle peut utiliser des horloges
▶ comme gardes sur les transitions
▶ comme invariants sur les états

Les horloges peuvent être
▶ testées dans les gardes
▶ (ré-)initialisées dans les mises à jour

K
3 ≥ x ≥ 2
x := 0

Les états

Un état (location) p d’un processus peut être :
▶ initial ;
▶ normal ;
▶ urgent (U) : les horloges ne peuvent pas progresser tant

qu’on reste en p, mais il est néanmoins possible d’y rester
et que globalement le système progresse (en temps nul) ;

▶ engagé (commit) (C) : patate chaude !
▶ Le système doit faire sortir immédiatement (en temps nul)

au moins un de ses processus se trouvant dans un état
engagé.

▶ Le temps ne peut donc pas progresser tant que le système
a un processus dans un état engagé.

Les transitions
Contiennent :
▶ sélection :

▶ liste de déclarations de la forme nom : type
▶ la portée de la variable est la transition
▶ elle peut masquer d’autres variables
▶ elle est initialisée aléatoirement
▶ a : int [0,3] : a prend une valeur entre 0 et 3
▶ types simples seulement

▶ gardes :
▶ condition pour que la transition soit éligible
▶ conjonction de

▶ conditions simples sur les horloges et différences d’horloges
▶ expressions booléennes sans horloges

▶ synchronisations : rendez-vous avec un autre processus
sur un canal
▶ c ! : « écriture », attente d’un lecteur sur c
▶ c? : « lecture », attente d’un écrivain sur c

Les transitions
Contiennent :
▶ mises à jour : liste d’expressions à effets de bord

▶ évaluées quand la transition est exécutée
▶ dans la liste, les expressions sont évaluées de gauche à

droite
▶ en cas de synchronisation, les mises à jour de l’écrivain

sont réalisées avant celles du lecteur
▶ poids : probabilité de choisir cette transition (pour les

modèles probabilistes)
Une transition t est éligible à l’exécution quand
▶ le processus est dans l’état source de t
▶ elle n’a pas de garde
▶ sa garde est satisfaite

Inutile que l’invariant de l’état destination q de t soit satisfait !
▶ il est possible que l’exécution de la mise à jour de t rende

l’invariant de q vrai

Les canaux

Ils
▶ servent en position de gardes sur les transitions
▶ ne transportent pas de données
▶ ne servent qu’à synchroniser les processus du système

entre eux (rendez-vous)
▶ c ! :

▶ émission sur c
▶ le processus exécutant attend sur le canal c un autre

processus exécutant c?
▶ c? : dual de c

▶ peuvent être urgents
▶ peuvent être de diffusion (broadcast)

Un canal normal

Les canaux urgents (urgent chan)

▶ quand une synchronisation est possible, le temps ne
s’écoule plus jusqu’à ce qu’elle soit réalisée

▶ cela ne signifie pas que si une synchronisation est
possible, l’état suivant du système est celui où la
synchronisation a été réalisée

▶ les gardes d’horloges ne sont pas autorisées sur la
transition

Les canaux de diffusion (broadcast chan)

▶ synchronisation 1 − n,
▶ n peut être nul : l’écriture est réalisée, même si aucun

lecteur n’est disponible
▶ si plusieurs lecteurs sont disponibles, ils reçoivent tous le

signal : le système fait progresser simultanément l’écrivain
et tous ces lecteurs.

Un canal de diffusion

Différence entre état urgent et engagé : exemple
P1 veut envoyer une valeur v à P2
Stockage de la valeur dans une variable partagée t
Personne ne doit modifier t pendant la transmission

Échec !

Différence entre état urgent et engagé : exemple
P1 veut envoyer une valeur v à P2
Stockage de la valeur dans une variable partagée t
Personne ne doit modifier t pendant la transmission

Succès

Différence entre état urgent et engagé : une solution
plus simple

P2 veut envoyer une valeur v à P1
Stockage de la valeur dans une variable partagée t
Personne ne doit modifier t pendant la transmission

Succès car les MAJ de l’écrivain sont réalisées avant celles du lecteurNicolas Bedon - Introduction au model-checking - 121/134



Exemple : des trains, un pont

Données :
▶ N voies ferrées, un seul pont
▶ une seule voie sur le pont
▶ temps de freinage tf
▶ temps de traversée du pont constant

Problème :
▶ éviter les collisions (pont=ressource critique)
▶ éviter les famines (le temps de traversée doit être garanti)

Exemple : des trains, un pont
Modélisation du système :
▶ un processus par train
▶ un processus « contrôleur », qui peut décider la mise en

attente (arrêt) des trains
▶ chaque train prévient le contrôleur ta − tf avant arrivée

prévue sur le pont en lui laissant ta − tf pour donner un
éventuel ordre d’arrêt

▶ à l’arrivée d’un nouveau train, le contrôleur le place dans
une file (FIFO)

▶ si la file n’était pas vide, ordre d’arrêt du train
▶ chaque train quittant le pont prévient le contrôleur
▶ qui donne l’ordre au train (s’il existe) le plus anciennement

dans la file de redémarrer
Détails implantatoires :
▶ ta − tf = 10
▶ la file est implantée par un tableau

Exemple : des trains, un pont

Exemple : des trains, un pont

Exemple : des trains, un pont

Exemple : des trains, un pont

Exemple : des trains, un pont
UPPAAL autorise la définition de fonctions (pas de récursion,
évidemment !)

Java PathFinder : une autre approche

Développé par la NASA
https://babelfish.arc.nasa.gov/trac/jpf/wiki/
WikiStart

Machine virtuelle Java écrite en Java

Elle parcourt tous les chemins d’exécution
▶ elle est capable de manipulations symboliques

(1 ≤ x ∧ x ≤ 10)
▶ elle repère les états équivalents

Un système d’observateurs permet la vérification de propriétés.

Conclusion

Il existe
▶ beaucoup d’autres logiques (CTL*, ATL, . . .)
▶ beaucoup d’autres d’outils (SMV, . . .)

Bien choisir son model-checker en fonction de ce qu’on
souhaite faire

La qualité du model-checking dépend de la qualité du modèle
▶ le modèle TS et la propriété ϕ doivent être justes !
▶ la taille du modèle |TS| doit rester raisonnable.

Complexité du model-checking :
LTL |TS| exp(|ϕ|)
CTL |TS||ϕ|

